NRL Report 6815

A Calculation of Geomagnetic Time

ROBERT R. MEIER

E. O. Hulburt Center for Space Research
Space Science Division

April 3, 1969

NAVAL RESEARCH LABORATORY
Washington, D.C.

CONTENTS

ABSTRACT

Transformations have been derived, using matrix methods, which convert geographic solar time and position to geomagnetic time and position in both the centered and eccentric dipole coordinate systems. A computer program has been set up to facilitate the use of the transformations.

PROBLEM STATUS

This is an interim report; work is continuing on the problem.

AUTHORIZATION

NRL Problem A01-19B
Project NASA S-45340

Manuscript submitted October 9, 1968.

A CALCULATION OF GEOMAGNETIC TIME

INTRODUCTION

The concept of local geomagnetic time has proved to be useful in the interpretation of auroral and other phenomena associated with the earth's geomagnetic field. The concept of geomagnetic time was first introduced by Vegard (1) in connection with auroral studies. Vegard considered an earth-centered geomagnetic coordinate system. In this centered dipole system the local geomagnetic time at an observing station is defined as the hour angle of the sun relative to the local magnetic meridian in the earth-centered geomagnetic coordinate system, whose north pole is near Thule, Greenland. Later Vestine (2) suggested that time based on the eccentric dipole is more useful for studying time-varying geomagnetic phenomena. For eccentric dipole time, the local geomagnetic time of a point P is defined as the hour angle between two planes — namely, the plane defined by the magnetic dipole axis of the earth and a point located at the center of the sun and the plane defined by the magnetic dipole axis of the earth and the geographic point P.

Several discussions of geomagnetic time in both the centered and eccentric dipole coordinate systems have been published (3,4). Previous derivations have often used methods which are less than straightforward. Therefore, it is useful to provide a discussion of geomagnetic time using an alternate technique (matrices) and a computer program to facilitate its employment.

CENTERED DIPOLE TIME

Converting from the geographic coordinate system to the centered geomagnetic system involves three rotations, since both coordinate systems have the same origin. The geographic system is shown in Fig. 1. $\hat{S}_g$ is the unit solar vector and is given in column matrix form by

$$\hat{S}_g = \begin{bmatrix} \cos(180 - \omega t)\cos\delta \\ \sin(180 - \omega t)\cos\delta \\ \sin\delta \end{bmatrix} \tag{1}$$

where δ is the solar declination (measured positive north of the equator), t is Greenwich time, and ω is 15 degrees/hour. The unit vector representing the direction of the observing station $\hat{P}_g$ is given by

$$\hat{P}_g = \begin{bmatrix} \cos\phi_g \cos\lambda_g \\ \cos\phi_g \sin\lambda_g \\ \sin\phi_g \end{bmatrix} \tag{2}$$

where ϕ_g is the latitude and λ_g is the longitude measured east of Greenwich.



Fig. 1 - Geographic coordinate system

The centered geomagnetic system is easily defined in terms of an Euler coordinate system. The geomagnetic polar axis $\hat{m}_g$ is about 11.7 degrees from the geographic north pole on the 291-degree geographic longitude meridian. The transformation matrix from the geographic to the centered geomagnetic system (5) is

$$A = \begin{bmatrix} \cos\Psi\cos\phi - \cos\theta\sin\phi\sin\Psi & \cos\Psi\sin\phi + \cos\theta\cos\phi\sin\Psi & \sin\Psi\sin\theta \\ -\sin\Psi\cos\phi - \cos\theta\sin\phi\cos\Psi & -\sin\Psi\sin\phi + \cos\theta\cos\phi\cos\Psi & \cos\Psi\sin\theta \\ \sin\theta\sin\phi & -\sin\theta\cos\phi & \cos\theta \end{bmatrix} \tag{3}$$

where θ is 11.7 degrees, ϕ is 21 degrees, and Ψ is -90 degrees, since the geomagnetic prime meridian is intersected by $\hat{m}$ and the geographic north pole (Fig. 2).

Then in geomagnetic coordinates, the unit vectors become

$$\hat{P}_m = \begin{bmatrix} P_{1m} \\ P_{2m} \\ P_{3m} \end{bmatrix} = A\hat{P}_g$$

and (4)



Fig. 2 - Centered dipole coordinate system

$$\hat{S}_m = \begin{bmatrix} S_{1m} \\ S_{2m} \\ S_{3m} \end{bmatrix} = A\hat{S}_g$$

The geomagnetic hour angle t_m can then be found from

$$t_m = \frac{\lambda_m}{\omega} + T_m , \tag{5}$$

where λ_m is the geomagnetic longitude of $\hat{P}_m$ and T_m is the hour angle of $\hat{S}_m$ relative to the prime meridian. The longitude is given by

$$\tan \lambda_m = \frac{P_{2m}}{P_{1m}} , \tag{6}$$

while T_m is obtained from

$$\tan (180 - \omega T_m) = \frac{S_{2m}}{S_{1m}} \tag{7}$$

The geomagnetic latitude of $\hat{P}_m$, ϕ_m, is given by

$$\cos (90 - \phi_m) = P_{3m} . \tag{8}$$

ECCENTRIC DIPOLE TIME

The eccentric dipole coordinate system is obtained by laterally displacing the centered dipole without changing its orientation. The eccentric dipole unit vector $\hat{E}$ is shown relative to the centered dipole unit vector $\hat{m}$ in Fig. 3. The vector defining the displacement between the two coordinate systems D_g is given (in geographic coordinates) by

$$D_g = |d| \begin{bmatrix} \cos 15.6 \cos 150.9 \\ \cos 15.6 \sin 150.9 \\ \sin 15.6 \end{bmatrix}, \tag{9}$$

where the magnitude of the displacement $|d|$ is 0.0685 earth radii (6).



Fig. 3 - Eccentric dipole coordinate system referenced to centered dipole system

Because the eccentric coordinate system is obtained by a translation of the centered system, the sun's orientation is the same in both systems (the sun is assumed to be infinitely distant). Therefore, only the position matrix must be translated into the new system to determine eccentric dipole time. In the centered system the position of the observing station is

$$P_E = P_m - D_m, \tag{10}$$

where

$$D_m = AD_g \tag{11}$$

and

$$P_m = |p|\hat{P}_m. \tag{12}$$

The absolute value of P, $|p|$ is the distance of the observing station from the center of the earth.

In analogy to centered time, the eccentric time is

$$t_E = \frac{\lambda_E}{\omega} + T_m . \tag{13}$$

The eccentric longitude λ_E is given by

$$\tan \lambda_E = \frac{P_{2E}}{P_{1E}}, \tag{14}$$

where P_{1E} and P_{2E} are components of the unit vector

$$\hat{P}_E = \frac{P_E}{|P_E|} . \tag{15}$$

In addition the eccentric dipole latitude ϕ_E is given by

$$\cos (90 - \phi_E) = P_{3E} . \tag{16}$$

A Fortran computer program is given in the Appendix which accepts the solar declination, Greenwich time, and geographic latitude and longitude of an observation station and produces time, latitude, and longitude in both the centered and the eccentric dipole systems. Several examples are presented using the OGO-IV satellite as the observing station.

CONCLUSIONS

Formulas have been derived using matrix methods to provide conversions from geographic time and position to time and position in two geomagnetic coordinate systems. A computer code is also provided which calculates time with respect to both the centered and the eccentric geomagnetic dipoles.

ACKNOWLEDGMENT

This work was supported in part by the National Science Foundation under Grant GP-6354. The author acknowledges several useful discussions with G. T. Hicks.

REFERENCES

1. Agy, V., "Geomagnetic Coordinates and Geomagnetic Time," NBS Report 8789, 1965

2. Goldstein, H., "Classical Mechanics," Reading, Massachusetts:Addison-Wesley, 1950

3. Parkinson, W.D., and Cleary, J., "The Eccentric Geomagnetic Dipole," Geophys. J. 1:346 (1958)

4. Simonow, G.V., "Geomagnetic Time," Geophys. J. 8:258-267 (1963)

5. Vegard, L., "Bericht uber die neueren Untersuchungen am Nordlicht," Jahrb. Radioaktivitat Elektronik 14:383-466 (1917)

6. Vestine, E.H., "Asymmetrical Characteristics of the Earth's Magnetic Disturbance-Field," Terrest. Magn. Atmosph. Elec. 43:261-282 (1938)

Appendix

```
      PROGRAM MAGTIME
C
C  THIS PROGRAM CALCULATES LOCAL TIME IN BOTH THE CENTERED AND
C   ECCENTRIC MAGNETIC DIPLOE SYSTEMS.  ALSO, LATITUDE AND LONGITUDE ARE
C   CALCULATED IN EACH SYSTEM.
C
    1 FORMAT (F10.5)
    2 FORMAT (I3)
    3 FORMAT (3F10.5)
    5 FORMAT (/*  SOLAR  DECLINATION = * F10.5 /// 8X*GEOGRAPHIC* 10X
     $*GEOMAGNETIC -- CENTERED*3X*GEOMAGNETIC --ECCENTRIC* / *   GMT*5X
     $*LAT.*3X
     $*LONG.* 6X*TIME*4X*LAT.* 3X*LONG.*6X*TIME*4X*LAT.*3X*LONG.* //)
    6 FORMAT(1X,3(F6.1,2F8.2,4X))
7     FORMAT(A8)
77    FORMAT(1H1,A8)
      DIMENSION S(3), DMAG(3),PE(3), P(3), D(3)
      DIMENSION A(3,3), SM(3), PM(3),DM(3)
C  ALT IS THE ALTITUDE (KM.) OF OBSERVATION POINT.  RE IS THE RADIUS OF
C   THE EARTH.  OM IS 15 DEGREES / HOUR.
      ALT = 100.
      RE = 6371.
      OM = 15.
      R = ALT + RE
C  THE ROTATION MATRIX. A(I,J) FROM GEOGRAPHIC COORDINATES TO CENTERED
C   GEOMAGNETIC COORDINATES IS NOW DEFINED.
      TH = 11.7
      PH = 21.0
      A(1,1) = COSD(TH)*SIND(PH)
      A(1,2) = -COSD(TH)*COSD(PH)
      A(1,3) = -SIND(TH)
      A(2,1) = COSD(PH)
      A(2,2) = SIND(PH)
      A(2,3) = 0.
      A(3,1) = SIND(TH)*SIND(PH)
      A(3,2) = -SIND(TH)*COSD(PH)
      A(3,3) = COSD(TH)
C  D(I) IS THE DISPLACEMENT VECTOR OF THE ORIGIN OF THE ECCENTRIC DIPOLE
C   SYSTEM (IN GEOGRAPHIC COORDINATES).
      DVAL = .0685*RE
      D(1) = COSD(150.9) * COSD(15.6)   *DVAL
      D(2) = SIND(150.9) * COSD(15.6)   *DVAL
      D(3) = SIND(15.6)    *DVAL
C  READ IN THE DATE, NUMBER OF POINTS FOR THAT DATE AND THE SOLAR
C   DECLINATION.
   12 READ (3,7) DATE
      WRITE(1,77)DATE
      READ (3,2) IMAX
      READ (3,1)  SUNDEC
      WRITE (1,5) SUNDEC
      DO 4 II=1,IMAX
C  NOW READ THE GREENWICH MEAN TIME FOR EACH POINT ALONG WITH THE
C   LATITUDE AND LONGITUDE (GEOGRAPHIC).
      READ (3,3) GMT, GLAT, GLON
C  CONVERT TIME TO FRACTIONS OF HOURS.
      CALL TCONV(GMT,T,C)
C  P(I) IS THE OBSERVATION POSITION VECTOR AND S(I) IS THE SOLAR
```

```
C    DIRECTION.
       P(1) = COSD(GLON)*COSD(GLAT)*R
       P(2) = SIND(GLON)*COSD(GLAT)*R
       P(3) = SIND(GLAT)   *R
       S(1) = COSD(180.-OM*T) * COSD(SUNDEC)
       S(2) = SIND(180.-OM*T) * COSD(SUNDEC)
       S(3) = SIND(SUNDEC)
C   ROTATE INTO CENTERED GEOMAGNETIC COORDINATES.
       DO 39 I=1,3
       DM(I)=0.
       SM(I)=0.
   39 PM(I)=0.
       DO 30  I=1,3
       DO 30  J=1,3
       DM(I)= A(I,J)*D(J)+ DM(I)
       SM(I) = A(I,J)*S(J)+SM(I)
   30 PM(I) = A(I,J)*P(J)+PM(I)
C   CALCULATE GEOMAGNETIC UNIVERSAL TIME, LONGITUDE, LATITUDE AND LOCAL
C    TIME.
       UTMAG = (180.-ATANQ(SM(1),SM(2))*57.296)/OM
       IF (UTMAG) 13,14,14
   13 UTMAG= UTMAG+24.
   14 CONTINUE
C   NOTE THAT ATANQ ACCEPTS THE X AND Y COMPONENTS AND RETURNS THE ANGLE
C    IN THE CORRECT QUADRANT.
       XMAGLON = ATANQ(PM(1),PM(2))*57.296
       XMAGLAT = 90.- ACOS(PM(3) /R)*57.296
       TIMMAG = XMAGLON/OM + UTMAG
       IF (TIMMAG-24.)  40,40,41
   41 TIMMAG = TIMMAG-24.
   40 CONTINUE
C   NOW DO DISPLACEMENT INTO ECCENTRIC SYSTEM.
       DO 10 I=1,3
   10 PE(I)= PM(I)- DM(I)
       PEVAL= SQRT( PE(1)*PE(1)+PE(2)*PE(2)+PE(3)*PE(3) )
       DO 11 I=1,3
   11 PE(I)= PE(I)/PEVAL
       ECCLON= ATANQ(PE(1), PE(2) )*57.296
       ECCLAT= 90.- ACOS(PE(3) )*57.296
       TIMECC = ECCLON/OM + UTMAG
       IF (TIMECC-24.)  440,440,441
  441 TIMECC = TIMECC-24.
  440 IF (TIMECC) 15,16,16
   15 TIMECC= TIMECC+24.
   16 CONTINUE
C   CONVERT TIME TO HOURS AND MINUTES
       CALL TCONV(TIMMAG,TIMMAG,1)
       CALL TCONV(TIMECC,TIMECC,1)
       CALL TCONV(T,  T,1)
       WRITE (1,6) T,GLAT,GLON,TIMMAG,XMAGLAT,XMAGLON,TIMECC,ECCLAT,
      $ECCLON
    4 CONTINUE
       GO TO 12
       END
```

```
      SUBROUTINE TCONV (T,TC,I)
C
C..THIS CONVERTS T(HOURS,MIN) TO TC(FRAC. OF HOURS) IF I=0 ...VICE VERSA IF I=1
C
      IF (I) 1,1,2
    1 N=T/100.
      TEMP=N
      TFRAC= (T-TEMP*100.)/ 60.
      TC= TEMP+ TFRAC
      RETURN
    2 N=T
      TEMP=N
      TFRAC= (T-TEMP)* 60.
      TC= TEMP*100. + TFRAC
      RETURN
      END
```

670927

SOLAR DECLINATION = -1.50000

	GEOGRAPHIC		GEOMAGNETIC -- CENTERED			GEOMAGNETIC --ECCENTRIC		
GMT	LAT.	LONG.	TIME	LAT.	LONG.	TIME	LAT.	LONG.
6.0	77.00	-154.80	1051.3	73.23	231.03	1106.3	76.75	234.78
15.8	68.10	-4.60	157.5	70.26	95.15	123.5	67.92	86.65
56.0	-75.20	10.40	2320.9	-69.59	46.05	2316.0	-66.27	44.83
105.4	-67.70	163.10	1355.2	-72.41	262.27	1436.6	-74.97	272.63
243.6	-66.10	139.20	1322.5	-75.39	229.38	1338.1	-79.15	233.27
323.4	82.70	170.50	1215.6	73.38	202.50	1156.6	76.83	197.75
422.0	-63.80	115.40	1216.6	-75.44	187.74	1136.5	-78.60	177.73
955.5	84.90	91.10	1657.2	73.41	173.92	1612.4	75.70	162.73
1132.5	82.80	48.40	1723.1	73.70	156.64	1629.8	74.96	143.30
1354.4	-60.10	166.70	222.6	-64.94	256.45	248.0	-67.79	262.79
1445.6	74.60	-18.50	1641.6	78.05	98.36	1547.5	75.69	84.82
1531.2	-56.30	141.50	147.2	-65.73	223.24	151.0	-69.47	224.18
1538.6	-83.40	170.20	843.9	-79.95	325.53	956.9	-78.36	343.79
1708.4	-54.00	116.60	104.8	-65.62	187.99	43.0	-68.78	182.52
1717.4	-85.80	179.90	1117.3	-79.09	338.83	1215.5	-76.89	353.36
1759.2	67.60	-72.00	1259.8	79.27	353.85	1345.3	76.25	5.22
1847.5	-58.90	92.80	22.1	-69.72	152.27	2338.7	-71.01	141.41
1[illegible]55.2	-85.80	164.70	1306.4	-80.87	341.41	1410.7	-78.47	357.48

Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY *(Corporate author)* Naval Research Laboratory Washington, D.C. 20390	2a. REPORT SECURITY CLASSIFICATION UNCLASSIFIED 2b. GROUP

3. REPORT TITLE

A CALCULATION OF GEOMAGNETIC TIME

4. DESCRIPTIVE NOTES *(Type of report and inclusive dates)*

Interim report; work is continuing on the problem.

5. AUTHOR(S) *(First name, middle initial, last name)*

R. R. Meier

6. REPORT DATE April 3, 1969	7a. TOTAL NO. OF PAGES 14	7b. NO. OF REFS 6

8a. CONTRACT OR GRANT NO. NRL Problem A01-19B b. PROJECT NO. NASA S-45340-G c. d.	9a. ORIGINATOR'S REPORT NUMBER(S) NRL Report 6815 9b. OTHER REPORT NO(S) *(Any other numbers that may be assigned this report)*

10. DISTRIBUTION STATEMENT

This document has been approved for public release and sale; its distribution is unlimited.

11. SUPPLEMENTARY NOTES Sponsored by National Aeronautics and Space Administration, Washington, D.C.	12. SPONSORING MILITARY ACTIVITY

13. ABSTRACT

Transformations have been derived, using matrix methods, which convert geographic solar time and position to geomagnetic time and position in both the centered and eccentric dipole coordinate systems. A computer program has been set up to facilitate the use of the transformations.

14. KEY WORDS	LINK A		LINK B		LINK C	
	ROLE	WT	ROLE	WT	ROLE	WT
Geomagnetic time						
Centered dipole coordinate system						
Eccentric dipole coordinate system						